
ABSTRACT   
 
Missing values in survey data may lead to biased parameter estimates that do not accurately 
represent the population. 
 
There are a variety of techniques to deal with survey missing data. For instance, weighting 
adjustments are used in some studies to compensate for biased estimators produced by survey 
non-response (1).  Some conventional imputation methods such as model-based imputation 
(mean, ratio and regression imputation) are still used to replace missing values with acceptable 
values (2).  
 
However, adjusted weights need to be created for each variable with item non-response.  This 
requires a great deal of time and effort (2). Conventional imputation methods focus on estimating 
mean parameter values and ignore the between-imputation component of variability. Such 
imputation methods result in smaller standard errors than if variability were taken into account, 
thus increasing the likelihood of statistical significance.  Instead of filling in a single value for 
each missing value in survey data, more advanced multiple imputation (MI) can replace each 
missing value with a set of plausible values that represent the uncertainty about the right value to 
impute. 
 
Methods for multiple imputation are described in detail by Dr. Allison in his book as well as the 
materials for his course on multiple imputation.  These methods do not describe how to 
incorporate the fact that many of our large scale data sets use complex clustered sampling 
designs. As such, simple random sample methods, even with sample weights, may be 
inappropriate. 
 
This paper will demonstrate and discuss how we use the procedures MI, MIanalysis and 
Surveylogistic in SAS(r) to perform analyses with these three cumulative logit models using 
complex survey data.  
 
 
INTRODUCTION 
  
Proc Surveylogistic fits linear logistic regression models for discrete response survey data by the 
maximum likelihood method and incorporates the sample design into the analysis. 
 
By default, an observation is excluded in the analysis if it has a missing value for any variable, 
including the following variables: Weight, STRATA and CLUSTER.  
 



Proc MI and Proc MIanalysis are multiple imputation procedures. When Proc MI is used, 
missing values are imputed multiple times to generate multiple complete data sets.  Each 
imputed data set can be analyzed by a SAS standard procedure to estimate a desired model. The 
results from the complete data sets are combined to produce inferential results by Proc 
MIanalysis. This method reflects the uncertainty about the predictions of the unknown missing 
values, and the resulting estimated variances of the parameter estimates will be unbiased. 
 
 
DATA SOURCE  
 
The data set is the Early Childhood Longitudinal Survey, Kindergarten Cohort - Third Grade 
Longitudinal Public Use data file. Almost 20,000 (19,684) children were included in the data. 
The children were initially interviewed in kindergarten during the 1998-99 school year (Wave 1 
and Wave 2). A 30% subsample of children was re-interviewed in the first grade (Wave 3) but all 
were followed up in the Spring of kindergarten (Wave 4) and the Spring of third grade (Wave 5).  
 
The purpose of this paper is to test three cumulative logit models to see how child financial 
support from and contact with a nonresidential biological father are associated with children’s 
physical health following the separation of their parents. 
 
A subset of 1,765 children who fit the following criteria was selected: the child was living with 
his/her biological mother and without his/her biological father in kindergarten (K); the child’s 
biological father was still alive when he/she was in the third grade.  
 
Dependent variables: 
health (Child’s health was reported by the mother at K) 
support (Degree of contact with child at K) 
contact (Degree of support to child at K) 
 
Independent variables: 
meduc2 (Maternal education levels at W2)                   
gender (Child’s sex)  
rblack (Race black) 
rhisp (Race hisp) 
rother  (Race other) 
lnincome (Logged, (income+1)) 
i2ins (Health insurance coverage at K)             
marital (Mother's marital status at K.) 
mempl2 (Maternal employment at K)                   
depres2 (Mother feels depressed at K)                
bwp (Birth weight in pounds)  
     
 
STEPS FOR MI 
 
1. Check data using PROC MI with option nimpute = 0. 

proc mi data = sasuser.missing 
(keep = health1 support2 contact2 meduc2 gender rblack rhisp rother  lnincome i2ins marital mempl2 depres2 bwp 
YRDADLIV ) 
Nimpute = 0; 
 
Before we perform logistic regressions, we need to check missing patterns using Proc MI with 
nimpute = 0. The variable Yrdadliv (Number of years biodad lived with child) is an auxiliary 
variable. It is only for imputation and not in the analytical model. Missing values were found as 
i2ins(1), marital(2), mempl2(17), depres2(34),  bwp(38), minaw(349) and Yrdadliv (8). There 
are no missing values on dependent variables. 
 
2. Impute data using PROC MI 
 
proc mi data = sasuser.missing out =sasuser.imped seed = 123 nimpute = 3; 
var health1 support2 contact2 meduc2 gender rblack rhisp rother  lnincome i2ins marital mempl2 depres2 bwp 
minaw c1_5fp0 Yrdadliv ; 
run;  
 

Three dependent variables, all predictors, an auxiliary variable, and a sample weight variable 
were included in the PROC MI procedure because leaving out the dependent variables would 
cause bias (3). 
 
The MCMC method in Proc MI, by default, uses a single chain to produce five imputations. It 
also completes 200 burn-in iterations before the first imputation and 100 iterations between 
imputations (4). The number of iterations above can be changed by the user.  
 
Proc MI was forced to produce the same results every time by the option Seed = n.  Three 
complete data sets were generated after the imputation process since the option of the number of 
imputed data sets was changed to 3 in our code. 
 
3. Analyze data using Proc Surveylogistic with by _imputation_; 
 
*run model  1 using proc surveylogistic; 
 proc surveylogistic data = sasuser.imped; 
stratum c15fpstr; 
cluster  c15fppsu; 
weight  c1_5fp0; 
model  health1(descending) = meduc2 gender rblack rhisp rother  lnincome i2ins marital depres2 bwp / link = 
clogit covb expb; *EXPB: displays the exponentiated coefficients (i.e., the odds ratios); 
   *COVB displays the covariance matrix of the parameter estimates); 
by _imputation_; 
ods output parameterestimates = model_1imp  covb = covmat1;  
run;  
 
*modify the SAS data of Model_1imp to avoid error message. Because variable of Intercept is not on the COVB = 
data set; 
data model_1imp; 
set model_1imp(rename = (variable = var)); 
if classval0 ne . then variable = var||'_'||classval0; 

else variable = var; 
drop var; 
run; 
 
*combine results into a data set of parameter estimates, standard errors and so on; 
proc mianalyze parms = model_1imp covb = covmat1; 
modeleffects intercept_4 intercept_3 intercept_2 meduc2 gender rblack rhisp rother lnincome i2ins marital  depres2  
bwp ; 
ods output parameterestimates = comb_health ;  
run; 

*************************************************************************************************; 

*run model 2  (cumulative logit model with a ordinal response variable ); 
proc surveylogistic data = sasuser.imped; 
stratum  c15fpstr; 
cluster  c15fppsu; 
weight  c1_5fp0; 
model    contact2(descending) =    
meduc2 gender rblack rhisp rother  lnincome i2ins marital  / link = clogit covb expb; 
by _imputation_; 
ods output parameterestimates = model_2imp  covb = covmat2;  
run;  
 
*modify the SAS data of Model_2imp to avoid error message. Because variable of Intercept is not on the COVB = 
data set; 
data model_2imp; 
set model_2imp(rename = (variable = var)); 
if classval0 ne . then variable = var||'_'||classval0; 
else variable = var; 
drop var; 
run; 
 
*combine results into a data set of parameter estimates, standard errors and so on; 
proc mianalyze parms = model_2imp covb = covmat2 mult;  
modeleffects intercept_3 intercept_2 meduc2 gender rblack rhisp rother  lnincome i2ins marital  minaw; 
ods output parameterestimates = comb_contact ;  
run; 
*************************************************************************************************; 

*run model 3 (generalized logits mode with a nominal response variable ); 
proc surveylogistic data = sasuser.imped; 
stratum  c15fpstr; 
cluster c15fppsu; 
weight c1_5fp0; 
model    support2  (descending) =    
meduc2 gender rblack rhisp rother  lnincome i2ins marital minaw mempl2/ link = clogit covb expb;  /*Specifies link 
function*/ 
by _imputation_; 
ods output parameterestimates = model_3imp  covb = covmat3;  
run;  
  
*modify the SAS data of Model_1imp to avoid error message. Because variable of Intercept is not on the COVB = 
data set; 
data model_3imp; 
set model_3imp(rename = (variable = var)); 
if classval0 ne . then variable = var||'_'||classval0; 

else variable = var; 
drop var; 
run; 
 
*combine results into a data set of parameter estimates, standard errors and so on; 
proc mianalyze parms = model_3imp covb = covmat3 mult;  
modeleffects intercept_2 intercept_1 meduc2 gender rblack rhisp rother  lnincome i2ins marital   mempl2; 
ods output parameterestimates = comb_support;  
run; 
 
ods output close; 
 
Analysis of three cumulative logit models was performed with an ordinal response variable in 
each model using the By _imputation_ statement in the Surveylogistic procedure. Variable 
Health of the 1st model is children’s health, which was reported by the mothers of the children 
while they were in kindergarten [K( 1 = Reg/Poor, 2 = Very Good, 3 = Good and 4 = Excellent)]; 
variable Contact of the 2nd model is Biodad Contact at W2 ( 1 = Contact during last month, 2 =  
Last contact between 1 year and 1 month, and 3 = No contact during last year ); and variable 
Support of the 3rd model is Biodad Supports at W2 ( 0 = No support, 1 = Awarded support & 2 = 
Reg support). After the above analysis was completed, two input data, i.e. parameter estimates 
listed in the PARMS =  and covariance matrices listed in the COVB =,  were used in Procedure 
MIanalyze. Output data set listed in ODS output was a combined result for model predictors. 
 
3. Display results 

Two data sets were finally combined together as the results of the analysis of cumulative logit 
models using data steps for each model, of which one data set was produced from unimputed 
data while the other from imputed data.  
 
data temp(keep  = model variable estimate p_value stderr df  lclmean uclmean waldchisq parm); 
retain model  "Original Data";  
set model_1 comb_1;  
if ProbChiSq = " " then p_value = round(Probt, 0.0001); 
else p_value = round(ProbChiSq,0.0001); 
if variable = " " then model  = "Imputed Data"  ; 
run; 
data sasuser.compare_health; 
retain model  variable estimate odds p_value; 
set temp;  
if variable = " " then variable = parm ; 
odds = exp(estimate) ; 
drop parm; 
run; 
 
 
CONCLUSION 
 
Multiple imputation can be used with any data and model. Proc MI and Proc MIanalysis are easy 
to learn and to use with SAS standard procedures such as Proc REG, Proc Logistic, and Proc 
GLM. Attention should be given to the fact that the data set listed in Covb = ,  which was created 
by the Surveylogistic procedure using the imputed data sets, may not include intercept variables. 
One needs to modify it in order to avoid an error message when Proc MIanalyze with option 
Covb = is used. 
 
The results from studies reported here, which were analyzed by Proc Surveylogistic with 
imputed data, make much more sense than those with unimputed data.  
 
 
TRADEMARKS  

